

Paper 286-2013

Gee! No, GTL! Visualizing Data With The SAS Graph Template Language

Ted Conway, Chicago, IL


ABSTRACT

When you need to produce a grid of related graphs with minimum coding, PROC SGPANEL is hard to beat. But eventually you'll run into a situation that demands more precise control over the output. Perhaps there are unusual scaling/formatting requirements. Or information needs to be presented in a specific order. Or things need to be clarified via annotations or other markup. That's where the Graph Template Language (GTL) can help. In this paper, we'll see how GTL can be used to create a customized grid of time series plots from segments and measures found in the TOTARRESTS sample data set. This may be of interest to all skill levels. It requires Base SAS, SAS GTL, and the SAS Macro Facility on UNIX or the PC.

SO, WHAT'S THE BIG IDEA?

In *The Graph Template Language: Beyond the SAS/GRAPH® Procedures*, a SAS Global Forum 2012 paper, Jesse M. Pratt describes why it's sometimes worth extra effort to create a more highly-customized data visualization using SAS's Graph Template Language (GTL) instead of employing the easier-to-use SGPLOT and SGPANEL.

This short paper presents an example of how the GTL and macro techniques presented by Pratt can be combined with custom SAS functions (PROC FCMP) and new SAS 9.3 GTL features (transparency) to create a simple-but-information-rich lattice chart that presents data from SAS's TOTARRESTS sample data set.



The devil is in the details (see **Appendix** for full code & output), but to summarize, to create the above image:

1. **PROC SGRENDER** was used to create a composite .png image containing the many charts.
2. **Layout Lattice** was used to define a grid of 30 charts – 5 rows of measures X 6 columns of age ranges.
3. The **seriesplot** statement was used to create each of the individual charts.
4. For each chart, **PROC FCMP** was used to run a DATA step to create macro variables containing the ending value for each measure, the Year-Over-Year change from the prior value (difference and %) , and an indicator of whether this represents an increase or decrease (the **unicode** value for an up or down arrow).
5. An **entry** statement was used to display a Yahoo Finance stock quote-like summary for each chart containing:
 - An **eye-catcher arrow** indicating whether the final value is an increase/decrease over the prior year
 - The **difference** between the final value and the prior year's value
 - The **percentage difference** between the ending value and the prior year's value
6. A **drawtext** statement was used to prominently display the ending value for each measure on the "wall" of the series plot. SAS 9.3's **transparency** option is used to prevent text from obscuring the plot lines and markers.

ABOUT THE DATA

Figure 2 - SAS Enterprise Guide TOTARRESTS Sample Data Set
http://support.sas.com/learn/statlibrary/statlib_eg4.2/data/totarrests.sas7bdat

Name	Obs	Vars	Variable	Type	Length	Format	Informat	Label
TOTARRESTS[1]	30	19	Year	Numeric	8	BEST12.	F12.	Year
			TotalArrests	Numeric	8	BEST12.	F12.	Total Number of Arrests
			AGE1	Numeric	8	BEST12.	F12.	Total Arrests for Ages 14 and Under
			AGE2	Numeric	8	BEST12.	F12.	Total Arrests for Ages 15-17
			AGE3	Numeric	8	BEST12.	F12.	Total Arrests for Ages 18-20
			AGE4	Numeric	8	BEST12.	F12.	Total Arrests for Ages 21-24
			AGE5	Numeric	8	BEST12.	F12.	Total Arrests for Ages 25 and Older
			ArrestRate	Numeric	8	BEST12.	F12.	Arrests per 100,000 population
			AGE1rate	Numeric	8	BEST12.	F12.	Arrests per 100,000 for ages 14 and under
			AGE2rate	Numeric	8	BEST12.	F12.	Arrests per 100,000 for ages 15-17
			AGE3rate	Numeric	8	BEST12.	F12.	Arrests per 100,000 for ages 18-20
			AGE4rate	Numeric	8	BEST12.	F12.	Arrests per 100,000 for ages 21-24
			AGE5rate	Numeric	8	BEST12.	F12.	Arrests per 100,000 for ages 25 and older
			Population	Numeric	8	BEST12.	F12.	Population
			AGE1pop	Numeric	8	BEST12.	F12.	AGE1pop
			AGE2pop	Numeric	8	BEST12.	F12.	AGE2pop
			AGE3pop	Numeric	8	BEST12.	F12.	AGE3pop
			AGE4pop	Numeric	8	BEST12.	F12.	AGE4pop
			AGE5pop	Numeric	8	BEST12.	F12.	AGE5pop

	Year	TotalArrests	AGE1	AGE5	ArrestRate	AGE1rate	AGE5rate	Population	AGE1pop	AGE5pop
1	1999	14031100	771710.5	7675011.7	5141.8949127	1320.7662291	4345.3464949	272878000	58429000	176626000
2	1998	14528300	828113.1	7932451.8	5374.8996482	1424.6126718	4533.9695695	270299000	58129000	174956000
3	1997	15284300	917058	8360512.1	5708.5499582	1582.8537895	4827.5035944	267744000	57937000	173185000
4	1996	15168100	925254.1	8312118.8	5763.4157741	1593.0957833	4913.5871274	263179000	58079000	169166000
5	1995	15119800	937427.6	8421728.6	5754.4214865	1623.9261338	4980.9135321	262751000	57726000	169080000
6	1994	14648700	966814.2	8012838.9	5626.4331394	1685.6080339	4797.2740662	260355000	57357000	167029000
7	1993	14036300	842178	7776110.2	5444.5202982	1484.0925511	4712.708375	257806000	56747000	165003000
8	1992	14075100	816355.8	7797605.4	5518.7597288	1458.7159603	4786.4792	255041000	55964000	162909000
9	1991	14211900	810078.3	7717061.7	5636.7573742	1470.2232345	4803.0208936	252129000	55099000	160671000

REFERENCES, ACKNOWLEDGEMENTS, RECOMMENDED READING

- √ Pratt, Jesse M. *The Graph Template Language: Beyond the SAS/GRAPH® Procedures.*
support.sas.com/resources/papers/proceedings12/285-2012.pdf
- √ SAS. *SAS(R) 9.3 Graph Template Language: Reference, Third Edition.*
support.sas.com/documentation/cdl/en/grstatgraph/65377/HTML/default/viewer.htm#p0891qx3y0z8xqn1k9ijhv5xughi.htm
- √ Matange, Sanjay. Heath, Dan. *Statistical Graphics Procedures by Example: Effective Graphs Using SAS.*
support.sas.com/pubscat/bookdetails.jsp?pc=63855.

CONTACT INFORMATION

Ted Conway resides in Chicago, Illinois. Spam filters notwithstanding, he can be reached at tedconway@aol.com.

TRADEMARKS

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

COMPLETE SAS CODE

```

libname l 'C:\Program Files\SAS\9.3\SASEnterpriseGuide\5.1\Sample\Data';
*==> Rename totarrests SAS dataset variables & compute percentages of totals;
%macro renameVars;
proc sql;
create table totarrests as
select Year, Population as Population0, TotalArrests as Arrests0, ArrestRate as ArrestRate0
%do i=1 %to 5;
, age&i.pop as Population&i, age&i as Arrests&i, age&i.rate as ArrestRate&i
%end;
%do i=0 %to 5;
, Population&i/Population0 as PctPopulation&i, Arrests&i/Arrests0 as PctArrests&i
%end;
from l.totarrests;
%mend;
%renameVars;

*==> Find ending value for specified var and YOY change (Absolute & % Differences);
%let lastvar=?; %let pctchg=?; %let delta=?; %let updown=?;
%macro getdelta;
%let yvar = %sysfunc(dequote(&yvar));
data _null_;
set totarrests nobs=n point=n;
format Population: Arrests: ArrestRate: commall. Pct: percentn7.1;
call symput("lastvar",compress(vvalue(&yvar)));
saveval=&yvar;
prev=n-1;
set totarrests point=prev;
pctchg=(saveval-&yvar)/&yvar;
call symput("pctchg",compress(put(pctchg,percentn7.1)));
&yvar=saveval-&yvar;
call symput("delta",compress(vvalue(&yvar)));
if &yvar=0 | &yvar=. then call symput("updown","0020"); * Space (no change);
else if &yvar<0 then call symput("updown","25BC"); * Down arrow;
else call symput("updown","25B2"); * Up arrow;
stop;
run;
%mend;

*==> Define function to permit %getdelta macro calls from PROC SGRENDER;
proc fcmp outlib = sasuser.ds.functions;
function GetDeltas(yvar $);
rc = run_macro('getdelta', yvar);
return(rc);
endsub;
run;

option cmplib = (sasuser.ds);

*==> Define/generate one chart (SeriesPlot - y-axis Metric X x-axis Year);
%macro CreateOneChart(YVAR);
cell;
cellheader;
%let rc=%sysfunc(getdeltas("&yvar"));
entry textattrs=(size=19pt) halight=left " " halight=right {unicode "&UPDOWN"x} "&delta.
(&pctchg.);
endcellheader;
layout overlay /
yaxisopts=(label=" " labelattrs=(color=black) tickvalueattrs=(size=14pt color=black))
xaxisopts=(label=" " labelattrs=(color=black) tickvalueattrs=(size=14pt color=black));
drawtext textattrs=(size=56pt color=black) "lastvar" /
width=100 drawspace=WALLPERCENT widthunit=percent
xspace=wallpercent yspace=wallpercent x=50 y=50 justify=center transparency=.80;
seriesplot x=Year y=&YVAR / lineattrs=(color=black thickness=1.5px) display=(markers)
MARKERATTRS=(SYMBOL=circlefilled color=black size=10px);
endlayout;
endcell;
%mend;

```

COMPLETE SAS CODE (CONTINUED)

```

*==> Define/generate grid of charts (Lattice - 5 Rows of Metrics X 6 Columns of Age Ranges);
%macro CreateAllCharts;
proc template;
%let Segments=1|2|3|4|5|0|;
%let Metrics=Population|PctPopulation|Arrests|PctArrests|ArrestRate|;
define statgraph mygraphs.example;
begingraph / designwidth=4050px designheight=2000px border=false;
layout lattice / backgroundcolor=white opaque=true columns=6 border=false;
entrytitle textattrs=(size=36pt weight=bold) "Arrest Statistics 1970-1999";
entrytitle textattrs=(size=24pt weight=bold) " ";
column2headers;
entry textattrs=(size=24pt weight=bold) "Ages 0-14";
entry textattrs=(size=24pt weight=bold) "Ages 15-17";
entry textattrs=(size=24pt weight=bold) "Ages 18-20";
entry textattrs=(size=24pt weight=bold) "Ages 21-24";
entry textattrs=(size=24pt weight=bold) "Ages 25+";
entry textattrs=(size=24pt weight=bold) "Total";
endcolumn2headers;
rowheaders;
entry textattrs=(size=24pt weight=bold) "Population" / rotate=90;
entry textattrs=(size=24pt weight=bold) "% Population" / rotate=90;
entry textattrs=(size=24pt weight=bold) "Arrests" / rotate=90;
entry textattrs=(size=24pt weight=bold) "% Arrests" / rotate=90;
entry textattrs=(size=24pt weight=bold) "Arrest Rate" / rotate=90;
endrowheaders;
sidebar / align=top; entry textattrs=(size=18pt) " " / valign=center; endsidebar;
%do r=1 %to 5;
  %do c=1 %to 6;
    %CreateOneChart(%scan(&Metrics, &r, '|') %scan(&Segments, &c, '|'));
  %end;
%end;
endlayout;
endgraph;
end;
run;

proc sgrender data=totarrests template="mygraphs.example";
format Population: Arrests: ArrestRate: commall. Pct: percentn7.4;
run;
%mend;

*==> Create file containing grid of charts (Arrests.png);
ods graphics on / antialias antialiasmax=10000 reset imagename="sampleGTLchart";
ods listing gpath="c:\temp";

%CreateAllCharts;
run;

```

OUTPUT (PNG IMAGE)

Figure 3 - GTL Lattice Chart

